

Clipping media using a script for macOS

ELAN supports the creation of media clips (video and/or audio) by external applications, based on information entered in a text file named "clip-media.txt". Via this script, ELAN can call an external program from the command line and pass the parameters it needs to do the actual clipping.

On **macOS** this works slightly different from how it works on **Windows** and **Linux** (see the file "Clip_media_Windows.pdf"). This tutorial shows you how to set-up this script on **macOS** to call an AppleScript script, which in turn calls QuickTime Player to:

1. open the media file
2. trim to the selection made in ELAN
3. export or save the result in the same format as the original or as .mov
4. close the file

The result will be saved in the same location as the original video or in a location specified by the user.

A few steps need to be taken to clip the media successfully:

1. Downloading and placing the script files in the right location.
2. Editing and saving 'clip-media.txt' to match your file paths and AppleScript choice.
3. Editing and saving the AppleScript of your choice (optional).
4. Calling the script from within ELAN via File->Export as->Media Clip using Script...

1. Where to get the script

The necessary files used to be part of the ELAN distribution but are not anymore (because of problems with digital signing of the application). The files can now be downloaded separately here:

https://www.mpi.nl/tools/elan/atemp/mac/clip_scripts_mac.zip

2. Where to place the script

Clipping with these AppleScripts will only work if you are working on a local machine.

There is now only one location where ELAN will find the main "script" file (i.e. "clip-media.txt") in a reliable manner. ELAN will look in the Preferences folder, which is located here:

```
/Users/<user>/Library/Preferences/ELAN
```

(<user> represents the home folder of the user, the user_login_name).

If the txt-file and the AppleScripts are located there, future versions of ELAN will also be able to find and use the scripts. In the Finder the <user>/Library/ folder is hidden by default; it can be made

visible via the `Go` menu of the Finder while holding down the `⌘` Option (or Alt) key.

After unzipping the downloaded zip file, copy the `.txt` and the `.scpt` files to the preferences folder as mentioned above.

The `.zip` contains the files:

- `"clip-media.txt"`
- `qtp_clip_10_10_export.scpt` : for macOS 10.10 or higher, the *export* function
- `qtp_clip_10_10_save.scpt` : for macOS 10.10 or higher, *save as* function (for `.mov`)
- `qtp_clip_12_2_export.scpt` : tested on macOS 12 (might work on older systems too)

This last *export* script attempts to overcome problems stemming from the user's setting for Decimal separator in Language & Region in System Preferences.

3. Editing of "clip-media.txt"

Working with the different AppleScripts requires some editing of `"clip-media.txt"`. When opening the file in a text-editor it will look something like this:

```
osascript /Users/<user>/Library/Preferences/ELAN/qtp_clip_10_10_export.scpt
$in_file $out_file $begin(sec.ms) $end(sec.ms)
```

```
# M2-edit-cl /in:$begin(fr) /out:$end(fr) $in_file $out_file
```

```
# ffmpeg.exe -vcodec copy -acodec copy -ss $begin(sec.ms) -t
"$duration(sec.ms)" -i $in_file $out_file
```

```
# Lines starting with a # (dash) are considered to be comment lines.
```

```
# ...
```

The first non-comment line, starting with `osascript`, is relevant for **macOS** users. `"osascript"` is a program that can execute AppleScript scripts (amongst others). It is called with the **absolute** path to the `.scpt` script-file, followed by four additional parameters. The placeholders (`$`) for these parameters are defined by ELAN; ELAN replaces the placeholders when calling the script. These parameters concern the start and end time of the selection and the input and output files. The placeholders should not be edited.

An example of an edited rule with an absolute script path:

```
osascript /Users/<user>/Library/Preferences/ELAN/qtp_clip_10_10_export.scpt
$in_file $out_file $begin(sec.ms) $end(sec.ms)
```

The script-file used in the example is `qtp_clip_10_10_export.scpt`: a script intended for **macOS** 10.10 or higher. This script leaves the extension intact; it only exports the selection to a new clip. The new clip can be found in the same folder as the original video unless the user has specified the

name and location in a *Save As...* window.

When you have edited the rule in `clip-media.txt` to your needs, save it and (re-)launch ELAN.

Note that on macOS it is also possible to use other applications for clipping (e.g. FFmpeg), similar to the procedure described in `Clip_media_Windows.pdf`.

4. Editing AppleScript files

As said, AppleScript is used to execute the QuickTime commands. These `.scpt` files also could require some editing. By double-clicking such a file, the Script Editor will open:

```
qtp_clip_10_10_export.sct
AppleScript <No selected element>

property hd : "Macintosh HD" -- change this if the hard drive has a different label

on run args
    set inFile to POSIX file (item 1 of args)
    set outFile to (item 2 of args) -- use as string first
    set start_string to (item 3 of args)
    set end_string to (item 4 of args)
    set start_time to (start_string as real)
    set end_time to (end_string as real)

    -- check the format of the start time
    set pindex to offset of "." in start_string
    if pindex > 0 then
        set start_length to length of start_string
        if start_length = pindex + 3 then
            set start_string to start_string & "1"
            -- set start_time to start_time - 1.0E-4
            try
                set start_time to (start_string as real)
            on error
                set start_string to (get text 1 thru (pindex - 1) of start_string) & "," & (get text (pindex + 1) thru
                    (start_length + 1) of start_string)
                set start_time to (start_string as real)
            end try
            -- display dialog "Start: " & start_string & " real: " & start_time
        end if
    end if

    -- convert posix path to HFS format, e.g. from /Users/Shared to disc:Users:Shared etc
    set AppleScript's text item delimiters to {"/"}
    set itemlist to get text items of outFile
    set AppleScript's text item delimiters to {":"}

    set target_file to hd & itemlist as string

    tell application "QuickTime Player"

        open inFile

        delay 1
        activate
        if not (exists document 1) then error number -128

        set the current time of document 1 to start_time
        trim document 1 from start_time to end_time

        delay 1
        --settings
        -- "720p"
        -- "iPad, iPhone 4 & Apple TV"
        export document 1 in file target_file using settings preset "720p" -- or "iPhone" or "Computer" etc.

        --delay 1
        close document 1 saving no
        set visible of window 1 to false
        -- quit saving no
    end tell
end run

Description
```

The first line defines a property for the name of the hard drive. The default is set to "Macintosh HD"

but this should be changed if the hard drive of the user has a different name. The name of the hard drive can be found e.g. by selecting `Go->Computer` in the Finder menu.

Another statement that might require editing is the one setting the output format:

```
export document 1 in file target_file using settings preset "720p"
```


Depending on the input file and the version of **macOS**, the following presets might be available:

- 480p
- 720p
- 1080p
- 4K
- Audio Only

The list of formats in the `Export As` menu of QuickTime Player (after opening your file in the player) can be used as a reference. It is advised to first create some clips in QuickTime Player itself, to get familiar with the possibilities and the limitations of clipping with QuickTime Player.

5. Executing the script in ELAN


After saving the script file(s), launch ELAN, open a document or create a new one, make a selection and from the main menu choose: `File->Export As->Media Clip using Script...`



QuickTime Player will start up and process the video. If you chose the "export" script, you might also see an export dialog. You can close that when finished.

The result will either be saved in the same location as the original video (the output file name will be constructed based on the input file name and the start and end time of the selection), or in a location specified by the user. This depends on a preference setting in the Media panel in

Edit->Preferences->Edit Preferences...



If the option Prompt for a clip file name is selected, a Save as prompt will be shown when the menu item of Figure 2 is selected. When using the export script, it depends on the type of the source file and on the output settings, whether to enter a file name with or without extension. If the source is e.g. an mpeg-1 file, adding the extension .mp4 is required in order to export a clip (which will be an mp4 file). If the input file is an .mp4 file, the extension can be omitted. When an .mp4 extension has been entered, a linked .wav file will also be clipped and saved as an audio only .mp4 file. If the output setting is "Audio Only", use the .m4a extension.